<!--#Include File="Include.asp"-->
<!--#Include File="../Class/Cls_Template.asp"--><%
'--------------------------------------
'Power By 方卡在线 @2010-2014 f02
'URL:http://www.fangka.net/
'EMail:admin@ifangka.com
'QQ:861118936
'版权所有 未经允许不得去除本版权信息
'--------------------------------------

Call CheckAdminType("0,1",UType)
Call CheckAdminLimit(ULimit,"m_Module",0)

Dim Fk_Module_Name,Fk_Module_SubTitle,Fk_Module_Keyword,Fk_Module_Description,Fk_Module_Pic,Fk_Module_Type,Fk_Module_FixedUrl,Fk_Module_Dir,Fk_Module_Url,Fk_Module_PageSize,Fk_Module_Model,Fk_Module_GBookReply,Fk_Module_IsIndex,Fk_Module_Menu,Fk_Module_Level,Fk_Module_Order,Fk_Module_LevelList,Fk_Module_Template,Fk_Module_LowTemplate,Fk_Module_LinkType,Fk_Module_Link,Fk_Module_State,Fk_Module_MenuShow,Fk_Module_Field,Fk_Module_SubColCount
Dim TemplateSelect
Dim Fk_Module_IsIndexShow

Types=Clng(Request.QueryString("Type"))

MenuId=Clng(Request.QueryString("Menu"))
Sqlstr="Select Fk_Menu_Name,Fk_Menu_Dir From [Fk_Menu] Where Fk_Menu_Id=" & MenuId
Rs.Open Sqlstr,Conn,1,1
If Not Rs.Eof Then
	MenuName=Rs("Fk_Menu_Name")
	MenuDir=Rs("Fk_Menu_Dir")
Else
	Rs.Close
	Call FKFun.ShowErr("菜单不存在！",1)
End If
Rs.Close

If MenuDir<>"" Then
	Sqlstr="Select Fk_Module_Id From [Fk_Module] Where Fk_Module_IsIndex=1 And Fk_Module_Menu=" & MenuId
	Rs.Open Sqlstr,Conn,1,1
	If Not Rs.Eof Then
		Fk_Module_IsIndexShow="0"
	Else
		Fk_Module_IsIndexShow="1"
	End If
	Rs.Close
Else
	Fk_Module_IsIndexShow="0"
End If

Select Case Types
	Case 1
		Call ModuleList() '菜单项列表
	Case 2
		Call ModuleAddForm() '添加菜单项表单
	Case 3
		Call ModuleAddDo() '执行添加菜单项
	Case 4
		Call ModuleEditForm() '修改菜单项表单
	Case 5
		Call ModuleEditDo() '执行修改菜单项
	Case 6
		Call ModuleDelDo() '执行删除菜单项
	Case 7
		Call ModuleOrderForm() '菜单项排序表单
	Case 8
		Call ModuleOrderDo() '执行菜单项排序

	Case 20
		Call ModuleBatAddForm() '添加菜单项表单
	Case 30
		Call ModuleBatAddDo() '执行添加菜单项
		
	Case Else
		Call FKFun.ShowErr("未发现功能项！",1)
End Select

'==========================================
'函 数 名：ModuleList()
'作	用：菜单项列表
'==========================================
Sub ModuleList()
	Set FKTemplate=New Cls_Template
%>
		<div id="rightNav">
			<ul>
				<li><a href="Module.asp?Type=2&Menu=<%=MenuId%>" target="dialog" width="950">添加菜单项</a></li>
				<li><a href="Module.asp?Type=7&Menu=<%=MenuId%>" target="get" div="mainRight">菜单项排序</a></li>
				<li><a href="Module.asp?Type=20&Menu=<%=MenuId%>" target="dialog" width="950">批量添加菜单项</a></li>
			</ul>
		</div>
		<div id="searchLine" class="font_bold">
			<%=MenuName%>——栏目管理
		</div>
		<table width="100%" bordercolor="#CCCCCC" border="1" cellspacing="0" cellpadding="0">
			<tr>
				<th align="center">ID</th>
				<th align="center">名称</th>
				<th align="center">链接</th>
				<th align="center">类型</th>
				<th align="center">模型</th>
				<th align="center">菜单显示</th>
				<th align="center">是否可用</th>
				<th align="center">操作</th>
			</tr>
<%
	Call GetModuleList(0,MenuId,0,"","")
%>
			<tr>
				<td height="30" colspan="20">&nbsp;</td>
			</tr>
		</table>
		<div id="rightBottom">
		
		</div>
<%
End Sub

'==========================================
'函 数 名：GetModuleList()
'作	用：菜单项列表
'==========================================
Function GetModuleList(gType,gMenu,gModule,gDefault,gTitle)
	Dim Rst,tryTitle,mUrl
	Set Rst=Server.Createobject("Adodb.RecordSet")
	If gTitle="" Then
		tryTitle="&nbsp;&nbsp;&nbsp;├"
	Else
		tryTitle="&nbsp;&nbsp;&nbsp;"&gTitle
	End If
	Sqlstr="Select Fk_Module_Id,Fk_Module_Name,Fk_Module_Type,Fk_Module_Model,Fk_Module_FixedUrl,Fk_Module_Url,Fk_Module_Link,Fk_Module_IsIndex,Fk_Module_MenuShow,Fk_Module_State,Fk_Module_Order From [Fk_Module] Where Fk_Module_Menu="&gMenu&" And Fk_Module_Level="&gModule&" Order By Fk_Module_Order Asc,Fk_Module_Id Asc"
	Rst.Open Sqlstr,Conn,1,1
	If Not Rst.Eof Then
		i=1
		While Not Rst.Eof
			If gType=0 Then
				If Instr("123",Rst("Fk_Module_Type"))>0 Then
					Fk_Module_Model=GetModelName(Rst("Fk_Module_Model"))
				Else
					Fk_Module_Model="-"
				End If
				If Rst("Fk_Module_IsIndex")=1 Then
					Fk_Module_IsIndex="&nbsp;&nbsp;<span style='color:#FF6633'>[首页]</span>"
				Else
					Fk_Module_IsIndex=""
				End If
				
				If Instr("0123",Rst("Fk_Module_Type"))>0 Then
					mUrl=FKTemplate.GetUrl(FullSiteDir,"",Rst("Fk_Module_Url"),"","",1,Rst("Fk_Module_Id"),Rst("Fk_Module_FixedUrl"))
					mUrl="<a href='"&mUrl&"' target='_blank'>"&mUrl&"</a>"
				ElseIf Instr("4",Rst("Fk_Module_Type"))>0 Then
					mUrl="转向到"&Rst("Fk_Module_Link")
				Else
					mUrl="-"
				End If
%>
			<tr>
				<td height="20" align="center"><%=Rst("Fk_Module_Id")%></td>
				<td>&nbsp;&nbsp;<%=gTitle%><%=Rst("Fk_Module_Name")%><%=Fk_Module_IsIndex%></td>
				<td>&nbsp;&nbsp;<%=mUrl%></td>
				<td align="center"><%=FKFun.GetModuleTypeName(Rst("Fk_Module_Type"))%></td>
				<td align="center"><%=Fk_Module_Model%></td>
				<td align="center"><%=FKFun.GetShowName(Rst("Fk_Module_MenuShow"))%></td>
				<td align="center"><%=FKFun.GetStateName(Rst("Fk_Module_State"))%></td>
				<td align="center"><a href="Module.asp?Type=4&Menu=<%=MenuId%>&Id=<%=Rst("Fk_Module_Id")%>" target="dialog" width="950">修改</a> <a href="Module.asp?Type=6&Menu=<%=MenuId%>&Id=<%=Rst("Fk_Module_Id")%>" target="delete" confirmStr="您确认要删除“<%=Rst("Fk_Module_Name")%>”？此操作不可逆！" refreshUrl="Module.asp?Type=1&Menu=<%=MenuId%>" showStr="名称删除成功！">删除</a></td>
			</tr>
<%
			ElseIf gType=1 Then
%>
					<option value="<%=Rst("Fk_Module_Id")%>"<%=FKFun.BeSelect(gDefault,Rst("Fk_Module_Id"))%>><%=gTitle%><%=Rst("Fk_Module_Name")%></option>
<%
			ElseIf gType=2 Then
				If Instr("123",Rst("Fk_Module_Type"))>0 Then
					Fk_Module_Model=GetModelName(Rst("Fk_Module_Model"))
				Else
					Fk_Module_Model="-"
				End If
				
				If Instr("0123",Rst("Fk_Module_Type"))>0 Then
					mUrl=FKTemplate.GetUrl(FullSiteDir,"",Rst("Fk_Module_Url"),"","",1,Rst("Fk_Module_Id"),Rst("Fk_Module_FixedUrl"))
					mUrl="<a href='"&mUrl&"' target='_blank'>"&mUrl&"</a>"
				ElseIf Instr("4",Rst("Fk_Module_Type"))>0 Then
					mUrl="转向到"&Rst("Fk_Module_Link")
				Else
					mUrl="-"
				End If
%>
			<tr>
				<td height="20" align="center"><%=Rst("Fk_Module_Id")%></td>
				<td>&nbsp;&nbsp;<%=gTitle%><%=Rst("Fk_Module_Name")%></td>
				<td>&nbsp;&nbsp;<%=mUrl%></td>
				<td align="center"><%=FKFun.GetModuleTypeName(Rst("Fk_Module_Type"))%></td>
				<td align="center"><%=Fk_Module_Model%></td>
				<td align="center"><%=FKFun.GetShowName(Rst("Fk_Module_MenuShow"))%></td>
				<td align="center"><%=FKFun.GetStateName(Rst("Fk_Module_State"))%></td>
				<td>&nbsp;&nbsp;<%=gTitle%><input name="Fk_Module_Order<%=Rst("Fk_Module_Id")%>" type="text" class="input" id="Fk_Module_Order<%=Rst("Fk_Module_Id")%>" value="<%=Rst("Fk_Module_Order")%>" size="5" /></td>
			</tr>
<%
			End If
			Call GetModuleList(gType,gMenu,Rst("Fk_Module_Id"),gDefault,tryTitle)
			Rst.MoveNext
			i=i+1
		Wend
	End If
	Rst.Close
End Function

'==========================================
'函 数 名：ModuleAddForm()
'作	用：添加名称表单
'==========================================
Sub ModuleAddForm()
	Sqlstr="Select Fk_Template_Name From [Fk_Template] Where Not (Fk_Template_Name Like 'm_%%' Or Fk_Template_Name Like '%%/m_%%') Order By Fk_Template_Id Asc"
	Rs.Open Sqlstr,Conn,1,1
	While Not Rs.Eof
		If Instr(Rs("Fk_Template_Name"),"/")>0 Then
			TempArr=Split(Rs("Fk_Template_Name"),"/")
			Temp=","&TempArr(UBound(TempArr))&","
		Else
			Temp=","&Rs("Fk_Template_Name")&","
		End If
		If Instr(DisTemplate,Temp)=0 Then
			If TemplateSelect="" Then
				TemplateSelect=Rs("Fk_Template_Name")
			Else
				TemplateSelect=TemplateSelect&",,"&Rs("Fk_Template_Name")
			End If
		End If
		Rs.MoveNext
	Wend
	Rs.Close
%>
<form id="ModuleAdd" name="ModuleAdd" method="post" target="ajax" action="Module.asp?Type=3&Menu=<%=MenuId%>" showStr="新菜单项添加成功！" refreshUrl="Module.asp?Type=1&Menu=<%=MenuId%>" boxClose="1">
<div id="title">添加新菜单项</div>
<div id="box">
	<table width="90%" border="1" bordercolor="#CCCCCC" align="center" cellpadding="0" cellspacing="0">
		<tr>
			<td width="24%" height="30" align="right">名称：</td>
			<td width="76%">&nbsp;<input name="Fk_Module_Name" type="text" class="input" id="Fk_Module_Name" /><span class="help" title="<p>菜单项名称，请输入1-50个字符（两个字符为一个汉字）。</p>"></span></td>
		</tr>
		<tr>
			<td height="30" align="right">副标题：</td>
			<td>&nbsp;<input name="Fk_Module_SubTitle" type="text" class="input" id="Fk_Module_SubTitle" /><span class="help" title="<p>菜单项副标题，请输入1-50个字符（1个汉字为2个字符）。</p>"></span></td>
		</tr>
		<tr>
			<td height="30" align="right">关键字：</td>
			<td>&nbsp;<input name="Fk_Module_Keyword" type="text" class="input" id="Fk_Module_Keyword" size="50" /><span class="help" title="<p>多个关键字用英文逗号隔开，用于页面meta的keywords，请输入1-255个字符（两个字符为一个汉字）。</p>"></span></td>
		</tr>
		<tr>
			<td height="30" align="right">描述：</td>
			<td>&nbsp;<input name="Fk_Module_Description" type="text" class="input" id="Fk_Module_Description" size="50" /><span class="help" title="<p>请输入菜单项的一段文字描述，用于页面meta的description，请输入1-255个字符（两个字符为一个汉字）。</p>"></span></td>
		</tr>
		<tr>
			<td height="30" align="right">图片：</td>
			<td>&nbsp;<input name="Fk_Module_Pic" type="text" class="input" id="Fk_Module_Pic" size="30" /> <input type="button" class="picList_btn btnUpload" name="btnPic" id="btnPic" value="上传图片" callback="2" inputId="Fk_Module_Pic" fileAction="Upload.asp?immediate=3" /><span class="help" title="<p>上传菜单图片</p>"></span></td>
		</tr>
		<tr class="dis">
			<td height="30" align="right">绝对路径：</td>
			<td>&nbsp;<input name="Fk_Module_FixedUrl" type="text" class="input" id="Fk_Module_FixedUrl" size="30" /> *设定后无法修改<span class="help" title="<p>绝对路径，请输入1-255个字符，设定后将忽略目录设置，内容模块和带分页的留言模块必须要是目录。</p>"></span></td>
		</tr>
		<tr>
			<td height="30" align="right">目录：</td>
			<td>&nbsp;<input name="Fk_Module_Dir" type="text" class="input" id="Fk_Module_Dir" size="30" /> *设定后无法修改<span class="help" title="<p>目录名，请输入1-50个字符，必须是字母或数字。</p>"></span></td>
		</tr>
		<tr>
			<td height="30" align="right">分级：</td>
			<td>&nbsp;<select class="input" name="Fk_Module_Level" id="Fk_Module_Level">
					<option value="0">一级菜单项</option>
<%
	Call GetModuleList(1,MenuId,0,"","")
%>
				</select><span class="help" title="<p>菜单项分级。</p>"></span>
			</td>
		</tr>
		<tr>
			<td height="30" align="right">类型：</td>
			<td>&nbsp;<select class="input" name="Fk_Module_Type" id="Fk_Module_Type" onchange="ModuleTypeChange($(this).val());">
					<option value="0">静态模块</option>
					<option value="1">内容模块</option>
					<option value="2">信息模块</option>
					<option value="3">留言模块</option>
					<option value="4">转向链接</option>
					<option value="5">中转模块</option>
				</select><span class="help" title="<p>菜单项类型。</p>"></span>
			</td>
		</tr>
		<tr class="m_Url dis">
			<td height="30" align="right">跳转模式：</td>
			<td>&nbsp;<input name="Fk_Module_LinkType" type="radio" class="input" id="Fk_Module_LinkType" value="0" checked="checked" />直接跳转
				<input name="Fk_Module_LinkType" type="radio" class="input" id="Fk_Module_LinkType" value="1" />页面跳转<span class="help" title="<p>设置跳转链接的方式，如果选择“直接跳转”，则会直接输出链接到页面，可以做外链；如选择“页面跳转”，会生成一个页面，并在通过页面内代码跳转。</p>"></span>
			</td>
		</tr>
		<tr class="m_Url dis">
			<td height="30" align="right">跳转链接：</td>
			<td>&nbsp;<input name="Fk_Module_Link" type="text" class="input" id="Fk_Module_Link" size="50" /> <span class="help" title="<p>设置一个链接，链接请输入1-255个字符，支持相对链接。</p>"></span></td>
		</tr>
		<tr id="m_Template">
			<td height="30" align="right">显示模板：</td>
			<td>&nbsp;<select class="input" name="Fk_Module_Template" id="Fk_Module_Template">
					<option value="">默认模板</option>
<%
	TempArr=Split(TemplateSelect,",,")
	For Each Temp In TempArr
%>
					<option value="<%=Temp%>"><%=Temp%></option>
<%
	Next
%>
				</select><span class="help" title="<p>选择显示模板。</p>"></span>
			</td>
		</tr>
		<tr class="m_List dis">
			<td height="30" align="right">子内容模板：</td>
			<td>&nbsp;<select class="input" name="Fk_Module_LowTemplate" id="Fk_Module_LowTemplate">
					<option value="">默认模板</option>
<%
	TempArr=Split(TemplateSelect,",,")
	For Each Temp In TempArr
%>
					<option value="<%=Temp%>"><%=Temp%></option>
<%
	Next
%>
				</select><span class="help" title="<p>选择子内容模板。</p>"></span>
			</td>
		</tr>
		<tr id="m_Model" class="dis">
			<td height="30" align="right">模型：</td>
			<td>&nbsp;<select class="input" name="Fk_Module_Model" id="Fk_Module_Model">
					<option value="0">请先选择类型</option>
				</select><span class="help" title="<p>菜单项类型。</p>"></span>
			</td>
		</tr>
		<tr id="m_GBookReply" class="dis">
			<td height="30" align="right">留言分页输出：</td>
			<td>&nbsp;<select class="input" name="Fk_Module_GBookReply" id="Fk_Module_GBookReply">
					<option value="0">所有记录</option>
					<option value="1">已回复记录</option>
					<option value="2">未回复记录</option>
					<option value="3">不在前台显示留言</option>
				</select><span class="help" title="<p>留言分页输出记录类型。</p>"></span>
			</td>
		</tr>
		<tr class="m_List dis">
			<td height="30" align="right">每页条数：</td>
			<td>&nbsp;<select class="input" name="Fk_Module_PageSize" id="Fk_Module_PageSize">
					<option value="0">按站点设置配置条数</option>
<%
	For i=1 To 50
%>
					<option value="<%=i%>"><%=i%>条</option>
<%
	Next
%>
				</select><span class="help" title="<p>菜单项类型。</p>"></span>
			</td>
		</tr>
<%
	Sqlstr="Select Fk_Field_Id,Fk_Field_Name,Fk_Field_Remark,Fk_Field_Label,Fk_Field_Help,Fk_Field_Type,Fk_Field_Option From [Fk_Field] Where Fk_Field_Use=2 Order By Fk_Field_Id Asc"
	Rs.Open Sqlstr,Conn,1,1
	While Not Rs.Eof
		Select Case Rs("Fk_Field_Type")
			Case 0
				Call FKFun.GetField_Input(Rs("Fk_Field_Name"),Rs("Fk_Field_Label"),Rs("Fk_Field_Help"),"")
			Case 1
				Call FKFun.GetField_Editor(Rs("Fk_Field_Name"),Rs("Fk_Field_Label"),Rs("Fk_Field_Help"),"")
			Case 2
				Call FKFun.GetField_Upload(Rs("Fk_Field_Name"),Rs("Fk_Field_Label"),Rs("Fk_Field_Help"),"")
			Case 3
				Call FKFun.GetField_Select(Rs("Fk_Field_Name"),Rs("Fk_Field_Label"),Rs("Fk_Field_Help"),Rs("Fk_Field_Option"),"")
		End Select
		Rs.MoveNext
	Wend
	Rs.Close
%>
		<tr>
			<td height="30" align="right">菜单是否显示：</td>
			<td>&nbsp;
				<label><input type="radio" name="Fk_Module_MenuShow" class="input" id="Fk_Module_MenuShow" value="0" />不显示</label>
				<label><input type="radio" name="Fk_Module_MenuShow" class="input" id="Fk_Module_MenuShow" value="1" checked="checked" />显示</label><span class="help" title="<p>配置该菜单项是否在菜单输出里显示。</p>"></span>
			</td>
		</tr>
		<tr>
			<td height="30" align="right">是否可用：</td>
			<td>&nbsp;
				<label><input type="radio" name="Fk_Module_State" class="input" id="Fk_Module_State" value="0" />不可用</label>
				<label><input type="radio" name="Fk_Module_State" class="input" id="Fk_Module_State" value="1" checked="checked" />可用</label> <span class="help" title="<p>配置该菜单项是否可用。</p>"></span>
			</td>
		</tr>
<%
	If Fk_Module_IsIndexShow="1" Then
%>
		<tr>
			<td height="30" align="right">菜单首页：</td>
			<td>&nbsp;
				<label><input type="radio" name="Fk_Module_IsIndex" class="input" id="Fk_Module_IsIndex" value="0" />否</label>
				<label><input type="radio" name="Fk_Module_IsIndex" class="input" id="Fk_Module_IsIndex" value="1" checked="checked" />是</label><span class="help" title="<p>菜单首页。</p>"></span>
			</td>
		</tr>
<%
	End If
%>
	</table>
</div>
<div id="bottom">
	<input type="submit" class="button" name="Enter" id="Enter" value="添 加" />
	<input type="button" onclick="dialogClose();" class="button" name="button" id="button" value="关 闭" />
</div>
</form>
<%
End Sub

'==============================
'函 数 名：ModuleAddDo
'作	用：执行添加
'==============================
Sub ModuleAddDo()
	Fk_Module_Name=FKFun.HTMLEncode(Trim(Request.Form("Fk_Module_Name")))
	Fk_Module_SubTitle=FKFun.HTMLEncode(Trim(Request.Form("Fk_Module_SubTitle")))
	Fk_Module_Keyword=FKFun.HTMLEncode(Trim(Request.Form("Fk_Module_Keyword")))
	Fk_Module_Description=FKFun.HTMLEncode(Trim(Request.Form("Fk_Module_Description")))
	Fk_Module_Pic=FKFun.HTMLEncode(Trim(Request.Form("Fk_Module_Pic")))
	Fk_Module_FixedUrl=FKFun.HTMLEncode(Trim(Request.Form("Fk_Module_FixedUrl")))
	Fk_Module_Dir=Server.URLEncode(FKFun.HTMLEncode(Trim(Request.Form("Fk_Module_Dir"))))
	Fk_Module_Template=FKFun.HTMLEncode(Trim(Request.Form("Fk_Module_Template")))
	Fk_Module_LowTemplate=FKFun.HTMLEncode(Trim(Request.Form("Fk_Module_LowTemplate")))
	Fk_Module_Link=FKFun.HTMLEncode(Trim(Request.Form("Fk_Module_Link")))

	Fk_Module_Type=Trim(Request.Form("Fk_Module_Type"))
	Fk_Module_PageSize=Trim(Request.Form("Fk_Module_PageSize"))
	Fk_Module_Level=Trim(Request.Form("Fk_Module_Level"))
	Fk_Module_Model=Trim(Request.Form("Fk_Module_Model"))
	Fk_Module_GBookReply=Trim(Request.Form("Fk_Module_GBookReply"))
	Fk_Module_State=Trim(Request.Form("Fk_Module_State"))
	Fk_Module_MenuShow=Trim(Request.Form("Fk_Module_MenuShow"))
	Fk_Module_LinkType=Trim(Request.Form("Fk_Module_LinkType"))
	Fk_Module_IsIndex=Trim(Request.Form("Fk_Module_IsIndex"))
	
	Call FKFun.JsonString(Fk_Module_Name,1,50,0,"请输入名称！","名称不能大于50个字符！")
	Call FKFun.JsonString(Fk_Module_SubTitle,1,50,2,"请输入副标题！","副标题不能大于50个字符！")
	Call FKFun.JsonString(Fk_Module_Keyword,1,255,2,"请输入关键字！","关键字不能大于255个字符！")
	Call FKFun.JsonString(Fk_Module_Description,1,255,2,"请输入描述！","描述不能大于255个字符！")
	Call FKFun.JsonString(Fk_Module_Pic,1,50,2,"请上传图片！","图片不能大于50个字符！")
	Call FKFun.JsonString(Fk_Module_FixedUrl,1,255,2,"请输入绝对路径！","绝对路径不能大于255个字符！")
	Call FKFun.JsonString(Fk_Module_Dir,1,50,2,"请输入目录名！","目录名不能大于50个字符！")
	Call FKFun.JsonNum(Fk_Module_Level,"请选择菜单项分级！")
	Call FKFun.JsonNum(Fk_Module_State,"请选择菜单项是否可用！")
	Call FKFun.JsonNum(Fk_Module_MenuShow,"请选择菜单项是否在菜单上显示！")
	Call FKFun.JsonNum(Fk_Module_Type,"请选择菜单项类型！")
	If Instr("123",Fk_Module_Type)>0 Then
		Call FKFun.JsonNum(Fk_Module_Model,"请选择菜单项模型！")
	Else
		Fk_Module_Model=0
	End If
	If Instr("13",Fk_Module_Type)>0 Then
		Call FKFun.JsonNum(Fk_Module_PageSize,"请选择每页条数！")
	Else
		Fk_Module_PageSize=0
	End If
	If Instr("3",Fk_Module_Type)>0 Then
		Call FKFun.JsonNum(Fk_Module_GBookReply,"请选择留言分页输出类型！")
	Else
		Fk_Module_GBookReply=0
	End If
	If Instr("4",Fk_Module_Type)>0 Then
		Call FKFun.JsonString(Fk_Module_Link,1,255,0,"请输入跳转链接！","跳转链接不能大于255个字符！")
		Call FKFun.JsonNum(Fk_Module_LinkType,"请选择跳转模式！")
	Else
		Fk_Module_Link=""
		Fk_Module_LinkType=0
	End If
	
	'绝对路径格式判断
	If Fk_Module_FixedUrl<>"" Then
		If Instr("1",Fk_Module_Type)>0 Then '判断必须是目录的配置
			If Right(Fk_Module_FixedUrl,1)<>"/" Or (Instr("3",Fk_Module_Type)>0 And Fk_Module_GBookReply=3) Then
				Call FKFun.JsonError("内容模块和显示分页记录的留言模块的绝对路径必须是目录，用/结束！")
			End If
		Else '判断允许为目录或文件的配置
			If Right(Fk_Module_FixedUrl,1)<>"/" And Instr(Fk_Module_FixedUrl,".")=0 Then
				Call FKFun.JsonError("绝对路径必须指定目录或文件名，目录用/结束，文件名必须带文件类型后缀！")
			End If
			'文件名安全判别
			If Instr(Fk_Module_FixedUrl,".")>0 Then
				If (Len(Fk_Module_FixedUrl)-Len(Replace(Fk_Module_FixedUrl,".","")))>1 Then
					Call FKFun.JsonError("绝对路径只能有一个“.”符号！")
				End If
				If Instr(",.html,",","&Right(Fk_Module_FixedUrl,5)&",")=0 And Instr(",.xml,.htm,",","&Right(Fk_Module_FixedUrl,4)&",")=0 And Instr(",.shtml,",","&Right(Fk_Module_FixedUrl,6)&",")=0 Then
					Call FKFun.JsonError("后缀只能是.html/.htm/.xml/.shtml中的一种！")
				End If
			End If
		End If
	End If
		
	'设定菜单首页
	If Fk_Module_IsIndexShow="1" Then
		Call FKFun.JsonNum(Fk_Module_IsIndex,"请选择是否首页！")
		If Fk_Module_IsIndex=1 And Fk_Module_Type<>0 Then
			Call FKFun.JsonError("菜单首页必须是静态模块！")
		End If
	Else
		Fk_Module_IsIndex=0
	End If
	
	'自定义字段
	Sqlstr="Select Fk_Field_Label,Fk_Field_Type From [Fk_Field] Where Fk_Field_Use=2 Order By Fk_Field_Id Asc"
	Rs.Open Sqlstr,Conn,1,1
	While Not Rs.Eof
		If Rs("Fk_Field_Type")="2" Then
			Temp=FKFun.EditorImgTry(Trim(Request.Form("Fk_Field_"&Rs("Fk_Field_Label")&"")))
			Temp=Replace(Temp,"'","&#39;")
		Else
			Temp=FKFun.HTMLEncode(Trim(Request.Form("Fk_Field_"&Rs("Fk_Field_Label")&"")))
		End If
		Temp=Replace(Temp,"_Field_Get:","&nbsp;_Field_Get")
		Temp=Replace(Temp,"Get_Field_Line:","&nbsp;Get_Field_Line")
		Fk_Module_Field=Fk_Module_Field&Rs("Fk_Field_Label")&"_Field_Get:"&Temp&"|Get_Field_Line|"
		Rs.MoveNext
	Wend
	Rs.Close
	
	'分级与链接读取
	If Fk_Module_Level>0 Then
		Sqlstr="Select Fk_Module_LevelList,Fk_Module_Url From [Fk_Module] Where Fk_Module_Id="&Fk_Module_Level&" And Fk_Module_Menu="&MenuId&""
		Rs.Open Sqlstr,Conn,1,1
		If Not Rs.Eof Then
			ModuleLevelList=Rs("Fk_Module_LevelList")
			ModuleUrl=Rs("Fk_Module_Url")
		End If
		Rs.Close
	Else
		ModuleLevelList="0"
		ModuleUrl=""
	End If
	If Fk_Module_Level=0 Then
		Fk_Module_LevelList="0"
	Else
		If ModuleLevelList="0" Then
			Fk_Module_LevelList=","&Fk_Module_Level&","
		Else
			Fk_Module_LevelList=ModuleLevelList&Fk_Module_Level&","
		End If
	End If
	If Fk_Module_Dir="" Then
		Fk_Module_Url=ModuleUrl
	Else
		If MenuDir<>"" Then
			Fk_Module_Url=ModuleUrl&MenuDir&"/"&Fk_Module_Dir&"/"
		Else
			Fk_Module_Url=ModuleUrl&Fk_Module_Dir&"/"
		End If
	End If
	
	'菜单首页链接
	If Fk_Module_IsIndex=1 Then
		Fk_Module_Url=MenuDir&"/"
	End If
	
	Sqlstr="Select Fk_Module_Id,Fk_Module_Name,Fk_Module_SubTitle,Fk_Module_Keyword,Fk_Module_Description,Fk_Module_Pic,Fk_Module_About,Fk_Module_Content,Fk_Module_Type,Fk_Module_FixedUrl,Fk_Module_Dir,Fk_Module_Url,Fk_Module_PageSize,Fk_Module_Model,Fk_Module_GBookReply,Fk_Module_IsIndex,Fk_Module_Menu,Fk_Module_Level,Fk_Module_Order,Fk_Module_LevelList,Fk_Module_Click,Fk_Module_Template,Fk_Module_LowTemplate,Fk_Module_LinkType,Fk_Module_Link,Fk_Module_State,Fk_Module_MenuShow,Fk_Module_Field,Fk_Module_ToHtml,Fk_Module_Admin,Fk_Module_Ip,Fk_Module_Time From [Fk_Module] Where Fk_Module_Name='"&Fk_Module_Name&"' And Fk_Module_Menu="&MenuId&" And Fk_Module_Level="&Fk_Module_Level&""
	Rs.Open Sqlstr,Conn,1,3
	If Rs.Eof Then
		Application.Lock()
		Rs.AddNew()
		Rs("Fk_Module_Name")=Fk_Module_Name
		Rs("Fk_Module_SubTitle")=Fk_Module_SubTitle
		Rs("Fk_Module_Keyword")=Fk_Module_Keyword
		Rs("Fk_Module_Description")=Fk_Module_Description
		Rs("Fk_Module_Pic")=Fk_Module_Pic
		Rs("Fk_Module_About")=""
		Rs("Fk_Module_Content")=""
		Rs("Fk_Module_Type")=Fk_Module_Type
		Rs("Fk_Module_FixedUrl")=Fk_Module_FixedUrl
		Rs("Fk_Module_Dir")=Fk_Module_Dir
		Rs("Fk_Module_Url")=Fk_Module_Url
		Rs("Fk_Module_PageSize")=Fk_Module_PageSize
		Rs("Fk_Module_Model")=Fk_Module_Model
		Rs("Fk_Module_GBookReply")=Fk_Module_GBookReply
		Rs("Fk_Module_IsIndex")=Fk_Module_IsIndex
		Rs("Fk_Module_Menu")=MenuId
		Rs("Fk_Module_Level")=Fk_Module_Level
		Rs("Fk_Module_Order")=0
		Rs("Fk_Module_LevelList")=Fk_Module_LevelList
		Rs("Fk_Module_Click")=0
		Rs("Fk_Module_Template")=Fk_Module_Template
		Rs("Fk_Module_LowTemplate")=Fk_Module_LowTemplate
		Rs("Fk_Module_LinkType")=Fk_Module_LinkType
		Rs("Fk_Module_Link")=Fk_Module_Link
		Rs("Fk_Module_State")=Fk_Module_State
		Rs("Fk_Module_MenuShow")=Fk_Module_MenuShow
		Rs("Fk_Module_Field")=Fk_Module_Field
		Rs("Fk_Module_ToHtml")=1
		Rs("Fk_Module_Admin")=UId
		Rs("Fk_Module_Ip")=Request.ServerVariables("REMOTE_ADDR")
		Rs("Fk_Module_Time")=Now()
		Rs.Update()
		If Fk_Module_Dir="" And Fk_Module_IsIndex=0 Then
			Rs.Close
			Sqlstr="Select Fk_Module_Id,Fk_Module_Url From [Fk_Module] Where Fk_Module_Name='"&Fk_Module_Name&"' And Fk_Module_Menu="&MenuId&""
			Rs.Open Sqlstr,Conn,1,3
			If Not Rs.Eof Then
				If MenuDir<>"" Then
					MenuDir=MenuDir&"/"
				End If
				Fk_Module_Url=Fk_Module_Url&MenuDir&"Col"&Rs("Fk_Module_Id")&"/"
				Fk_Module_Url=Replace(Fk_Module_Url,"//","/")
				Rs("Fk_Module_Url")=Fk_Module_Url
				Rs.Update()
			End If
		End If
		If Fk_Module_Level>0 Then
			Sqlstr="Update [Fk_Module] Set Fk_Module_SubColCount=Fk_Module_SubColCount+1 Where Fk_Module_Id=" & Fk_Module_Level
			Conn.Execute(Sqlstr)
			If Fk_Site_Html=2 Then
				Sqlstr="Update [Fk_Module] Set Fk_Module_ToHtml=1 Where Fk_Module_Id In (0"&Fk_Module_LevelList&"0)"
				Conn.Execute(Sqlstr)
			End If
		End If
		Application.UnLock()
		Call FKFun.JsonSuccess()
	Else
		Rs.Close
		Call FKFun.JsonError("该菜单项已存在！")
	End If
	Rs.Close
End Sub

'==========================================
'函 数 名：ModuleEditForm()
'作	用：修改名称表单
'==========================================
Sub ModuleEditForm()
	Id=Clng(Request.QueryString("Id"))
	Sqlstr="Select Fk_Module_Id,Fk_Module_Name,Fk_Module_SubTitle,Fk_Module_Keyword,Fk_Module_Description,Fk_Module_Pic,Fk_Module_Type,Fk_Module_FixedUrl,Fk_Module_Dir,Fk_Module_PageSize,Fk_Module_Model,Fk_Module_GBookReply,Fk_Module_IsIndex,Fk_Module_Menu,Fk_Module_Level,Fk_Module_Order,Fk_Module_Template,Fk_Module_LowTemplate,Fk_Module_LinkType,Fk_Module_Link,Fk_Module_State,Fk_Module_MenuShow,Fk_Module_Field From [Fk_Module] Where Fk_Module_Id=" & Id
	Rs.Open Sqlstr,Conn,1,1
	If Not Rs.Eof Then
		Fk_Module_Name=Rs("Fk_Module_Name")
		Fk_Module_SubTitle=Rs("Fk_Module_SubTitle")
		Fk_Module_Keyword=Rs("Fk_Module_Keyword")
		Fk_Module_Description=Rs("Fk_Module_Description")
		Fk_Module_Pic=Rs("Fk_Module_Pic")
		Fk_Module_Type=Rs("Fk_Module_Type")
		Fk_Module_FixedUrl=Rs("Fk_Module_FixedUrl")
		Fk_Module_Dir=Rs("Fk_Module_Dir")
		Fk_Module_PageSize=Rs("Fk_Module_PageSize")
		Fk_Module_Model=Rs("Fk_Module_Model")
		Fk_Module_GBookReply=Rs("Fk_Module_GBookReply")
		Fk_Module_IsIndex=Rs("Fk_Module_IsIndex")
		Fk_Module_Menu=Rs("Fk_Module_Menu")
		Fk_Module_Level=Rs("Fk_Module_Level")
		Fk_Module_Order=Rs("Fk_Module_Order")
		Fk_Module_Template=Rs("Fk_Module_Template")
		Fk_Module_LowTemplate=Rs("Fk_Module_LowTemplate")
		Fk_Module_LinkType=Rs("Fk_Module_LinkType")
		Fk_Module_Link=Rs("Fk_Module_Link")
		Fk_Module_State=Rs("Fk_Module_State")
		Fk_Module_MenuShow=Rs("Fk_Module_MenuShow")
		Fk_Module_Field=Rs("Fk_Module_Field")
	Else
		Rs.Close
		Call FKFun.ShowErr("菜单项不存在！",1)
	End If
	Rs.Close
	
	Sqlstr="Select Fk_Template_Name From [Fk_Template] Where Not (Fk_Template_Name Like 'm_%%' Or Fk_Template_Name Like '%%/m_%%') Order By Fk_Template_Id Asc"
	Rs.Open Sqlstr,Conn,1,1
	While Not Rs.Eof
		If Instr(Rs("Fk_Template_Name"),"/")>0 Then
			TempArr=Split(Rs("Fk_Template_Name"),"/")
			Temp=","&TempArr(UBound(TempArr))&","
		Else
			Temp=","&Rs("Fk_Template_Name")&","
		End If
		If Instr(DisTemplate,Temp)=0 Then
			If TemplateSelect="" Then
				TemplateSelect=Rs("Fk_Template_Name")
			Else
				TemplateSelect=TemplateSelect&",,"&Rs("Fk_Template_Name")
			End If
		End If
		Rs.MoveNext
	Wend
	Rs.Close
%>
<form id="ModuleEdit" name="ModuleEdit" method="post" target="ajax" action="Module.asp?Type=5&Menu=<%=MenuId%>" showStr="菜单项修改成功！" refreshUrl="Module.asp?Type=1&Menu=<%=MenuId%>" boxClose="1">
<div id="title">修改菜单项</div>
<div id="box">
	<table width="90%" border="1" bordercolor="#CCCCCC" align="center" cellpadding="0" cellspacing="0">
		<tr>
			<td width="24%" height="30" align="right">名称：</td>
			<td width="76%">&nbsp;<input name="Fk_Module_Name" type="text" class="input" id="Fk_Module_Name" value="<%=Fk_Module_Name%>" /><span class="help" title="<p>菜单项名称，请输入1-50个字符（两个字符为一个汉字）。</p>"></span></td>
		</tr>
		<tr>
			<td height="30" align="right">副标题：</td>
			<td>&nbsp;<input name="Fk_Module_SubTitle" type="text" class="input" id="Fk_Module_SubTitle" value="<%=Fk_Module_SubTitle%>" /><span class="help" title="<p>菜单项副标题，请输入1-50个字符（1个汉字为2个字符）。</p>"></span></td>
		</tr>
		<tr>
			<td height="30" align="right">关键字：</td>
			<td>&nbsp;<input name="Fk_Module_Keyword" type="text" class="input" id="Fk_Module_Keyword" value="<%=Fk_Module_Keyword%>" size="50" /><span class="help" title="<p>多个关键字用英文逗号隔开，用于页面meta的keywords，请输入1-255个字符（两个字符为一个汉字）。</p>"></span></td>
		</tr>
		<tr>
			<td height="30" align="right">描述：</td>
			<td>&nbsp;<input name="Fk_Module_Description" type="text" class="input" id="Fk_Module_Description" value="<%=Fk_Module_Description%>" size="50" /><span class="help" title="<p>请输入菜单项的一段文字描述，用于页面meta的description，请输入1-255个字符（两个字符为一个汉字）。</p>"></span></td>
		</tr>
		<tr>
			<td height="30" align="right">图片：</td>
			<td>&nbsp;<input name="Fk_Module_Pic" type="text" class="input" id="Fk_Module_Pic" value="<%=Fk_Module_Pic%>" size="30" /> <input type="button" class="picList_btn btnUpload" name="btnPic" id="btnPic" value="上传图片" callback="2" inputId="Fk_Module_Pic" fileAction="Upload.asp?immediate=3" /><span class="help" title="<p>上传菜单图片</p>"></span></td>
		</tr>
		<tr>
			<td height="30" align="right">绝对路径：</td>
			<td>&nbsp;<input name="Fk_Module_FixedUrl" type="text" class="input" id="Fk_Module_FixedUrl" value="<%=Fk_Module_FixedUrl%>" size="30"<%If Fk_Module_FixedUrl<>"" Then%> readonly="readonly"<%End If%> /> *设定后无法修改<span class="help" title="<p>绝对路径，请输入1-255个字符，设定后将忽略目录设置，内容模块和带分页的留言模块必须要是目录。</p>"></span></td>
		</tr>
		<tr>
			<td height="30" align="right">目录：</td>
			<td>&nbsp;<input name="Fk_Module_Dir" type="text" class="input" id="Fk_Module_Dir" value="<%=Fk_Module_Dir%>" size="30"<%If Fk_Module_Dir<>"" Then%> readonly="readonly"<%End If%> /> *设定后无法修改<span class="help" title="<p>目录名，请输入1-50个字符，必须是字母或数字。</p>"></span></td>
		</tr>
		<tr>
			<td height="30" align="right">分级：</td>
			<td>&nbsp;<select class="input" name="Fk_Module_Level" id="Fk_Module_Level">
					<option value="0"<%=FKFun.BeSelect(Fk_Module_Level,0)%>>一级菜单项</option>
<%
	Call GetModuleList(1,MenuId,0,Fk_Module_Level,"")
%>
				</select><span class="help" title="<p>菜单项分级。</p>"></span>
			</td>
		</tr>
		<tr>
			<td height="30" align="right">类型：</td>
			<td>&nbsp;<select class="input" name="Fk_Module_Type" id="Fk_Module_Type" onchange="ModuleTypeChange($(this).val());">
					<option value="0"<%=FKFun.BeSelect(Fk_Module_Type,0)%>>静态模块</option>
					<option value="1"<%=FKFun.BeSelect(Fk_Module_Type,1)%>>内容模块</option>
					<option value="2"<%=FKFun.BeSelect(Fk_Module_Type,2)%>>信息模块</option>
					<option value="3"<%=FKFun.BeSelect(Fk_Module_Type,3)%>>留言模块</option>
					<option value="4"<%=FKFun.BeSelect(Fk_Module_Type,4)%>>转向链接</option>
					<option value="5"<%=FKFun.BeSelect(Fk_Module_Type,5)%>>中转模块</option>
				</select><span class="help" title="<p>菜单项类型。</p>"></span>
			</td>
		</tr>
		<tr class="m_Url dis">
			<td height="30" align="right">跳转模式：</td>
			<td>&nbsp;<input name="Fk_Module_LinkType" type="radio" class="input" id="Fk_Module_LinkType" value="0"<%=FKFun.BeCheck(Fk_Module_LinkType,0)%> />直接跳转
				<input name="Fk_Module_LinkType" type="radio" class="input" id="Fk_Module_LinkType" value="1"<%=FKFun.BeCheck(Fk_Module_LinkType,1)%> />页面跳转<span class="help" title="<p>设置跳转链接的方式，如果选择“直接跳转”，则会直接输出链接到页面，可以做外链；如选择“页面跳转”，会生成一个页面，并在通过页面内代码跳转。</p>"></span>
			</td>
		</tr>
		<tr class="m_Url dis">
			<td height="30" align="right">跳转链接：</td>
			<td>&nbsp;<input name="Fk_Module_Link" type="text" class="input" id="Fk_Module_Link" value="<%=Fk_Module_Link%>" size="30" /> *设定后无法修改<span class="help" title="<p>设置一个链接，链接请输入1-255个字符，支持相对链接。</p>"></span></td>
		</tr>
		<tr id="m_Template">
			<td height="30" align="right">显示模板：</td>
			<td>&nbsp;<select class="input" name="Fk_Module_Template" id="Fk_Module_Template">
					<option value="">默认模板</option>
<%
	TempArr=Split(TemplateSelect,",,")
	For Each Temp In TempArr
%>
					<option value="<%=Temp%>"<%=FKFun.BeSelect(Fk_Module_Template,Temp)%>><%=Temp%></option>
<%
	Next
%>
				</select><span class="help" title="<p>选择显示模板。</p>"></span>
			</td>
		</tr>
		<tr class="m_List dis" id="m_LowTemplate">
			<td height="30" align="right">子内容模板：</td>
			<td>&nbsp;<select class="input" name="Fk_Module_LowTemplate" id="Fk_Module_LowTemplate">
					<option value="">默认模板</option>
<%
	TempArr=Split(TemplateSelect,",,")
	For Each Temp In TempArr
%>
					<option value="<%=Temp%>"<%=FKFun.BeSelect(Fk_Module_LowTemplate,Temp)%>><%=Temp%></option>
<%
	Next
%>
				</select><span class="help" title="<p>选择子内容模板。</p>"></span>
			</td>
		</tr>
		<tr id="m_Model" class="dis">
			<td height="30" align="right">模型：</td>
			<td>&nbsp;<select class="input" name="Fk_Module_Model" id="Fk_Module_Model">
<%
	If Instr("123",Fk_Module_Type)>0 Then
		Select Case Fk_Module_Type
			Case 1
				Temp=1
			Case 2
				Temp=2
			Case 3
				Temp=3
			Case Else
				Temp=0
		End Select
		Sqlstr="Select Fk_Model_Id,Fk_Model_Name From [Fk_Model] Where Fk_Model_Type="&Temp&" And Fk_Model_State=1 Order By Fk_Model_Id Desc"
		Response.Write(Sqlstr)
		Rs.Open Sqlstr,Conn,1,1
		While Not Rs.Eof
%>
					<option value="<%=Rs("Fk_Model_Id")%>"<%=FKFun.BeSelect(Fk_Module_Model,Rs("Fk_Model_Id"))%>><%=Rs("Fk_Model_Name")%></option>
<%
			Rs.MoveNext
		Wend
		Rs.Close
	Else
%>
					<option value="0">请先选择类型</option>
<%
	End If
%>
				</select><span class="help" title="<p>菜单项类型。</p>"></span>
			</td>
		</tr>
		<tr id="m_GBookReply" class="dis">
			<td height="30" align="right">留言分页输出：</td>
			<td>&nbsp;<select class="input" name="Fk_Module_GBookReply" id="Fk_Module_GBookReply">
					<option value="0"<%=FKFun.BeSelect(Fk_Module_GBookReply,0)%>>所有记录</option>
					<option value="1"<%=FKFun.BeSelect(Fk_Module_GBookReply,1)%>>已回复记录</option>
					<option value="2"<%=FKFun.BeSelect(Fk_Module_GBookReply,2)%>>未回复记录</option>
				</select><span class="help" title="<p>留言分页输出记录类型。</p>"></span>
			</td>
		</tr>
		<tr class="m_List dis">
			<td height="30" align="right">每页条数：</td>
			<td>&nbsp;<select class="input" name="Fk_Module_PageSize" id="Fk_Module_PageSize">
					<option value="0"<%=FKFun.BeSelect(Fk_Module_PageSize,0)%>>按站点设置配置条数</option>
<%
	For i=1 To 50
%>
					<option value="<%=i%>"<%=FKFun.BeSelect(Fk_Module_PageSize,i)%>><%=i%>条</option>
<%
	Next
%>
				</select><span class="help" title="<p>菜单项类型。</p>"></span>
			</td>
		</tr>
<%
	Sqlstr="Select Fk_Field_Id,Fk_Field_Name,Fk_Field_Remark,Fk_Field_Label,Fk_Field_Help,Fk_Field_Type,Fk_Field_Option From [Fk_Field] Where Fk_Field_Use=2 Order By Fk_Field_Id Asc"
	Rs.Open Sqlstr,Conn,1,1
	While Not Rs.Eof
		Temp=FKFun.GetSub(Fk_Module_Field,Rs("Fk_Field_Label")&"_Field_Get:","|Get_Field_Line|","")
		Select Case Rs("Fk_Field_Type")
			Case 0
				Call FKFun.GetField_Input(Rs("Fk_Field_Name"),Rs("Fk_Field_Label"),Rs("Fk_Field_Help"),Temp)
			Case 1
				Call FKFun.GetField_Editor(Rs("Fk_Field_Name"),Rs("Fk_Field_Label"),Rs("Fk_Field_Help"),Temp)
			Case 2
				Call FKFun.GetField_Upload(Rs("Fk_Field_Name"),Rs("Fk_Field_Label"),Rs("Fk_Field_Help"),Temp)
			Case 3
				Call FKFun.GetField_Select(Rs("Fk_Field_Name"),Rs("Fk_Field_Label"),Rs("Fk_Field_Help"),Rs("Fk_Field_Option"),Temp)
		End Select
		Rs.MoveNext
	Wend
	Rs.Close
%>
		<tr>
			<td height="30" align="right">菜单是否显示：</td>
			<td>&nbsp;
				<label><input type="radio" name="Fk_Module_MenuShow" class="input" id="Fk_Module_MenuShow" value="0"<%=FKFun.BeCheck(Fk_Module_MenuShow,0)%> />不显示</label>
				<label><input type="radio" name="Fk_Module_MenuShow" class="input" id="Fk_Module_MenuShow" value="1"<%=FKFun.BeCheck(Fk_Module_MenuShow,1)%> />显示</label><span class="help" title="<p>配置该菜单项是否在菜单输出里显示。</p>"></span>
			</td>
		</tr>
		<tr>
			<td height="30" align="right">是否可用：</td>
			<td>&nbsp;
				<label><input type="radio" name="Fk_Module_State" class="input" id="Fk_Module_State" value="0"<%=FKFun.BeCheck(Fk_Module_State,0)%> />不可用</label>
				<label><input type="radio" name="Fk_Module_State" class="input" id="Fk_Module_State" value="1"<%=FKFun.BeCheck(Fk_Module_State,1)%> />可用</label><span class="help" title="<p>配置该菜单项是否可用。</p>"></span>
			</td>
		</tr>
<%
	If Fk_Module_IsIndex=1 Or Fk_Module_IsIndexShow="1" Then
%>
		<tr>
			<td height="30" align="right">菜单首页：</td>
			<td>&nbsp;<input type="radio" name="Fk_Module_IsIndex" class="input" id="Fk_Module_IsIndex" value="0"<%=FKFun.BeCheck(Fk_Module_IsIndex,0)%> />否
				<input type="radio" name="Fk_Module_IsIndex" class="input" id="Fk_Module_IsIndex" value="1"<%=FKFun.BeCheck(Fk_Module_IsIndex,1)%> />是<span class="help" title="<p>菜单首页。</p>"></span>
			</td>
		</tr>
<%
	End If
%>
	</table>
</div>
<div id="bottom">
	<input type="hidden" name="Id" value="<%=Id%>" />
	<input type="submit" class="button" name="Enter" id="Enter" value="修 改" />
	<input type="button" onclick="dialogClose();" class="button" name="button" id="button" value="关 闭" />
</div>
</form>
<script type="text/javascript">
$(document).ready(function(){
<%
	If Instr("123",Fk_Module_Type)>0 Then
%>
	$('#m_Model').show();
<%
	End If
	If Instr("13",Fk_Module_Type)>0 Then
%>
	$('.m_List').show();
<%
	End If
	If Instr("3",Fk_Module_Type)>0 Then
%>
	$('#m_GBookReply').show();
<%
	End If
	If Instr("4",Fk_Module_Type)>0 Then
%>
	$('#m_Template').hide();
	$('.m_Url').show();
<%
	End If
	If Instr("5",Fk_Module_Type)>0 Then
%>
	$('#m_Template').hide();
<%
	End If
%>
});
</script>
<%
End Sub

'==============================
'函 数 名：ModuleEditDo
'作	用：执行修改
'==============================
Sub ModuleEditDo()
	Fk_Module_Name=FKFun.HTMLEncode(Trim(Request.Form("Fk_Module_Name")))
	Fk_Module_SubTitle=FKFun.HTMLEncode(Trim(Request.Form("Fk_Module_SubTitle")))
	Fk_Module_Keyword=FKFun.HTMLEncode(Trim(Request.Form("Fk_Module_Keyword")))
	Fk_Module_Description=FKFun.HTMLEncode(Trim(Request.Form("Fk_Module_Description")))
	Fk_Module_Pic=FKFun.HTMLEncode(Trim(Request.Form("Fk_Module_Pic")))
	Fk_Module_FixedUrl=FKFun.HTMLEncode(Trim(Request.Form("Fk_Module_FixedUrl")))
	Fk_Module_Dir=Server.URLEncode(FKFun.HTMLEncode(Trim(Request.Form("Fk_Module_Dir"))))
	Fk_Module_Template=FKFun.HTMLEncode(Trim(Request.Form("Fk_Module_Template")))
	Fk_Module_LowTemplate=FKFun.HTMLEncode(Trim(Request.Form("Fk_Module_LowTemplate")))
	Fk_Module_Link=FKFun.HTMLEncode(Trim(Request.Form("Fk_Module_Link")))

	Fk_Module_Type=Trim(Request.Form("Fk_Module_Type"))
	Fk_Module_PageSize=Trim(Request.Form("Fk_Module_PageSize"))
	Fk_Module_Level=Trim(Request.Form("Fk_Module_Level"))
	Fk_Module_Model=Trim(Request.Form("Fk_Module_Model"))
	Fk_Module_GBookReply=Trim(Request.Form("Fk_Module_GBookReply"))
	Fk_Module_State=Trim(Request.Form("Fk_Module_State"))
	Fk_Module_MenuShow=Trim(Request.Form("Fk_Module_MenuShow"))
	Fk_Module_LinkType=Trim(Request.Form("Fk_Module_LinkType"))
	Fk_Module_IsIndex=Trim(Request.Form("Fk_Module_IsIndex"))
	
	Id=Trim(Request.Form("Id"))
	Call FKFun.JsonString(Fk_Module_Name,1,50,0,"请输入名称！","名称不能大于50个字符！")
	Call FKFun.JsonString(Fk_Module_SubTitle,1,50,2,"请输入副标题！","副标题不能大于50个字符！")
	Call FKFun.JsonString(Fk_Module_Keyword,1,255,2,"请输入关键字！","关键字不能大于255个字符！")
	Call FKFun.JsonString(Fk_Module_Description,1,255,2,"请输入描述！","描述不能大于255个字符！")
	Call FKFun.JsonString(Fk_Module_Pic,1,50,2,"请上传图片！","图片不能大于50个字符！")
	Call FKFun.JsonString(Fk_Module_FixedUrl,1,255,2,"请输入绝对路径！","绝对路径不能大于255个字符！")
	Call FKFun.JsonString(Fk_Module_Dir,1,50,2,"请输入目录名！","目录名不能大于50个字符！")
	Call FKFun.JsonNum(Fk_Module_Level,"请选择菜单项分级！")
	Call FKFun.JsonNum(Fk_Module_State,"请选择菜单项是否可用！")
	Call FKFun.JsonNum(Fk_Module_MenuShow,"请选择菜单项是否在菜单上显示！")
	Call FKFun.JsonNum(Fk_Module_Type,"请选择菜单项类型！")
	If Instr("123",Fk_Module_Type)>0 Then
		Call FKFun.JsonNum(Fk_Module_Model,"请选择菜单项模型！")
	Else
		Fk_Module_Model=0
	End If
	If Instr("13",Fk_Module_Type)>0 Then
		Call FKFun.JsonNum(Fk_Module_PageSize,"请选择每页条数！")
	Else
		Fk_Module_PageSize=0
	End If
	If Instr("3",Fk_Module_Type)>0 Then
		Call FKFun.JsonNum(Fk_Module_GBookReply,"请选择留言分页输出类型！")
	Else
		Fk_Module_GBookReply=0
	End If
	If Instr("4",Fk_Module_Type)>0 Then
		Call FKFun.JsonString(Fk_Module_Link,1,255,0,"请输入跳转链接！","跳转链接不能大于255个字符！")
		Call FKFun.JsonNum(Fk_Module_LinkType,"请选择跳转模式！")
	Else
		Fk_Module_Link=""
		Fk_Module_LinkType=0
	End If
	
	'绝对路径格式判断
	If Fk_Module_FixedUrl<>"" Then
		If Instr("1",Fk_Module_Type)>0 Then '判断必须是目录的配置
			If Right(Fk_Module_FixedUrl,1)<>"/" Or (Instr("3",Fk_Module_Type)>0 And Fk_Module_GBookReply=3) Then
				Call FKFun.JsonError("内容模块和显示分页记录的留言模块的绝对路径必须是目录，用/结束！")
			End If
		Else '判断允许为目录或文件的配置
			If Right(Fk_Module_FixedUrl,1)<>"/" And Instr(Fk_Module_FixedUrl,".")=0 Then
				Call FKFun.JsonError("绝对路径必须指定目录或文件名，目录用/结束，文件名必须带文件类型后缀！")
			End If
			'文件名安全判别
			If Instr(Fk_Module_FixedUrl,".")>0 Then
				If (Len(Fk_Module_FixedUrl)-Len(Replace(Fk_Module_FixedUrl,".","")))>1 Then
					Call FKFun.JsonError("绝对路径只能有一个“.”符号！")
				End If
				If Instr(",.html,",","&Right(Fk_Module_FixedUrl,5)&",")=0 And Instr(",.xml,.htm,",","&Right(Fk_Module_FixedUrl,4)&",")=0 And Instr(",.shtml,",","&Right(Fk_Module_FixedUrl,6)&",")=0 Then
					Call FKFun.JsonError("后缀只能是.html/.htm/.xml/.shtml中的一种！")
				End If
			End If
		End If
	End If

	Call FKFun.JsonNum(Id,"系统参数错误，请刷新页面！")
	
	'设定菜单首页
	If Fk_Module_IsIndexShow="1" Or Fk_Module_IsIndex<>"" Then
		Call FKFun.JsonNum(Fk_Module_IsIndex,"请选择是否首页！")
		If Fk_Module_IsIndex=1 And Fk_Module_Type<>0 Then
			Call FKFun.JsonError("菜单首页必须是静态模块！")
		End If
	Else
		Fk_Module_IsIndex=0
	End If
	
	'自定义字段
	Sqlstr="Select Fk_Field_Label,Fk_Field_Type From [Fk_Field] Where Fk_Field_Use=2 Order By Fk_Field_Id Asc"
	Rs.Open Sqlstr,Conn,1,1
	While Not Rs.Eof
		If Rs("Fk_Field_Type")="2" Then
			Temp=FKFun.EditorImgTry(Trim(Request.Form("Fk_Field_"&Rs("Fk_Field_Label")&"")))
			Temp=Replace(Temp,"'","&#39;")
		Else
			Temp=FKFun.HTMLEncode(Trim(Request.Form("Fk_Field_"&Rs("Fk_Field_Label")&"")))
		End If
		Temp=Replace(Temp,"_Field_Get:","&nbsp;_Field_Get")
		Temp=Replace(Temp,"Get_Field_Line:","&nbsp;Get_Field_Line")
		Fk_Module_Field=Fk_Module_Field&Rs("Fk_Field_Label")&"_Field_Get:"&Temp&"|Get_Field_Line|"
		Rs.MoveNext
	Wend
	Rs.Close
	
	'分级与链接读取
	If Fk_Module_Level>0 Then
		Sqlstr="Select Fk_Module_LevelList,Fk_Module_Url From [Fk_Module] Where Fk_Module_Id="&Fk_Module_Level&" And Fk_Module_Menu="&MenuId&""
		Rs.Open Sqlstr,Conn,1,1
		If Not Rs.Eof Then
			ModuleLevelList=Rs("Fk_Module_LevelList")
			ModuleUrl=Rs("Fk_Module_Url")
		End If
		Rs.Close
	Else
		ModuleLevelList="0"
		ModuleUrl=""
	End If
	If Fk_Module_Level=0 Then
		Fk_Module_LevelList="0"
	Else
		If ModuleLevelList="0" Then
			Fk_Module_LevelList=","&Fk_Module_Level&","
		Else
			Fk_Module_LevelList=ModuleLevelList&Fk_Module_Level&","
		End If
	End If
	If Fk_Module_Dir="" Then
		If MenuDir<>"" Then
			Fk_Module_Url=ModuleUrl&MenuDir&"/"&"Col"&Id&"/"
		Else
			Fk_Module_Url=ModuleUrl&"Col"&Id&"/"
		End If
	Else
		If MenuDir<>"" Then
			Fk_Module_Url=ModuleUrl&MenuDir&"/"&Fk_Module_Dir&"/"
		Else
			Fk_Module_Url=ModuleUrl&Fk_Module_Dir&"/"
		End If
	End If
	Fk_Module_Url=Replace(Fk_Module_Url,"//","/")
	If Fk_Module_Level=Id Or Instr(Fk_Module_LevelList,","&Id&",")>0 Then
		Call FKFun.JsonError("不能成为自己的子分类！")
	End If
	
	'菜单首页链接
	If Fk_Module_IsIndex=1 Then
		Fk_Module_Url=MenuDir&"/"
	End If
	
	Sqlstr="Select Fk_Module_Id,Fk_Module_Name,Fk_Module_SubTitle,Fk_Module_Keyword,Fk_Module_Description,Fk_Module_Pic,Fk_Module_Type,Fk_Module_FixedUrl,Fk_Module_Dir,Fk_Module_Url,Fk_Module_PageSize,Fk_Module_Model,Fk_Module_GBookReply,Fk_Module_IsIndex,Fk_Module_Level,Fk_Module_LevelList,Fk_Module_Template,Fk_Module_LowTemplate,Fk_Module_LinkType,Fk_Module_Link,Fk_Module_State,Fk_Module_MenuShow,Fk_Module_Field,Fk_Module_ToHtml From [Fk_Module] Where Fk_Module_Id=" & Id
	Rs.Open Sqlstr,Conn,1,3
	If Not Rs.Eof Then
		Application.Lock()
		Temp=Rs("Fk_Module_Level")
		Rs("Fk_Module_Name")=Fk_Module_Name
		Rs("Fk_Module_SubTitle")=Fk_Module_SubTitle
		Rs("Fk_Module_Keyword")=Fk_Module_Keyword
		Rs("Fk_Module_Description")=Fk_Module_Description
		Rs("Fk_Module_Pic")=Fk_Module_Pic
		Rs("Fk_Module_Type")=Fk_Module_Type
		'只有填入绝对路径是更改
		If Fk_Module_FixedUrl<>Rs("Fk_Module_FixedUrl") Then
			Rs("Fk_Module_FixedUrl")=Fk_Module_FixedUrl
		End If
		'只有当填入目录名的时候变换网址
		If Fk_Module_Dir<>Rs("Fk_Module_Dir") Then
			Rs("Fk_Module_Dir")=Fk_Module_Dir
			Rs("Fk_Module_Url")=Fk_Module_Url
		End If
		Rs("Fk_Module_PageSize")=Fk_Module_PageSize
		Rs("Fk_Module_Model")=Fk_Module_Model
		Rs("Fk_Module_GBookReply")=Fk_Module_GBookReply
		Rs("Fk_Module_IsIndex")=Fk_Module_IsIndex
		Rs("Fk_Module_Level")=Fk_Module_Level
		Rs("Fk_Module_LevelList")=Fk_Module_LevelList
		Rs("Fk_Module_Template")=Fk_Module_Template
		Rs("Fk_Module_LowTemplate")=Fk_Module_LowTemplate
		Rs("Fk_Module_LinkType")=Fk_Module_LinkType
		Rs("Fk_Module_Link")=Fk_Module_Link
		Rs("Fk_Module_State")=Fk_Module_State
		Rs("Fk_Module_MenuShow")=Fk_Module_MenuShow
		Rs("Fk_Module_Field")=Fk_Module_Field
		Rs("Fk_Module_ToHtml")=1
		Rs.Update()
		'子栏目数量计算
		If Fk_Module_Level>0 Then
			Rs.Close
			Sqlstr="Select Count(Fk_Module_Id) From [Fk_Module] Where Fk_Module_Level=" & Fk_Module_Level
			Rs.Open Sqlstr,Conn,1,1
			If Rs(0)>0 Then
				Fk_Module_SubColCount=Rs(0)
			Else
				Fk_Module_SubColCount=0
			End If
			Sqlstr="Update [Fk_Module] Set Fk_Module_SubColCount="&Fk_Module_SubColCount&" Where Fk_Module_Id=" & Fk_Module_Level
			Conn.Execute(Sqlstr)
			'变换父栏目，原父栏目的子栏目数量计算
			If Temp<>Fk_Module_Level Then
				Rs.Close
				Sqlstr="Select Count(Fk_Module_Id) From [Fk_Module] Where Fk_Module_Level=" & Temp
				Rs.Open Sqlstr,Conn,1,1
				If Rs(0)>0 Then
					Fk_Module_SubColCount=Rs(0)
				Else
					Fk_Module_SubColCount=0
				End If
				Sqlstr="Update [Fk_Module] Set Fk_Module_SubColCount="&Fk_Module_SubColCount&" Where Fk_Module_Id=" & Temp
				Conn.Execute(Sqlstr)
			End If
			If Fk_Site_Html=2 Then
				Sqlstr="Update [Fk_Module] Set Fk_Module_ToHtml=1 Where Fk_Module_Id In (0"&Fk_Module_LevelList&"0)"
				Conn.Execute(Sqlstr)
			End If
		End If
		Application.UnLock()
		Call FKFun.JsonSuccess()
	Else
		Rs.Close
		Call FKFun.JsonError("要修改的菜单项不存在！")
	End If
	Rs.Close
End Sub

'==============================
'函 数 名：ModuleDelDo
'作	用：执行删除
'==============================
Sub ModuleDelDo()
	Id=Trim(Request.QueryString("Id"))
	Call FKFun.ShowNum(Id,"系统参数错误，请刷新页面！")
	Sqlstr="Select Fk_View_Id From [Fk_View] Where Fk_View_Module=" & Id
	Rs.Open Sqlstr,Conn,1,1
	If Not Rs.Eof Then
		Call FKFun.ShowErr("该菜单项在使用中，无法删除！",2)
	End If
	Rs.Close
	Sqlstr="Select Fk_GBook_Id From [Fk_GBook] Where Fk_GBook_Module=" & Id
	Rs.Open Sqlstr,Conn,1,1
	If Not Rs.Eof Then
		Call FKFun.ShowErr("该菜单项在使用中，无法删除！",2)
	End If
	Rs.Close
	Sqlstr="Select Fk_Module_Id From [Fk_Module] Where Fk_Module_Level=" & Id
	Rs.Open Sqlstr,Conn,1,1
	If Not Rs.Eof Then
		Call FKFun.ShowErr("该菜单项在使用中，无法删除！",2)
	End If
	Rs.Close
	Sqlstr="Select Fk_Module_Id From [Fk_Module] Where Fk_Module_Id=" & Id
	Rs.Open Sqlstr,Conn,1,3
	If Not Rs.Eof Then
		Application.Lock()
		Rs.Delete()
		Application.UnLock()
		Response.Write("ok")
	Else
		Response.Write("要删除的菜单项不存在！")
	End If
	Rs.Close
End Sub

'==========================================
'函 数 名：ModuleOrderForm()
'作	用：菜单项排序表单
'==========================================
Sub ModuleOrderForm()
	Set FKTemplate=New Cls_Template
%>
		<div id="rightNav">
			<ul>
				<li><a href="Module.asp?Type=1&Menu=<%=MenuId%>" target="get" div="mainRight">返回</a></li>
			</ul>
		</div>
		<div id="searchLine" class="font_bold">
			<%=MenuName%>——菜单项排序
		</div>
<%
'修复菜单项排序无反应的bug
%>
		<form id="ModuleOrder" name="ModuleOrder" method="post" target="ajax" action="Module.asp?Type=8&Menu=<%=MenuId%>" showStr="菜单项修改成功！" refreshUrl="Module.asp?Type=7&Menu=<%=MenuId%>">
		<table width="100%" bordercolor="#CCCCCC" border="1" cellspacing="0" cellpadding="0">
			<tr>
				<th align="center">ID</th>
				<th align="center">名称</th>
				<th align="center">链接</th>
				<th align="center">类型</th>
				<th align="center">模型</th>
				<th align="center">菜单显示</th>
				<th align="center">是否可用</th>
				<th align="center">排序</th>
			</tr>
<%
	Call GetModuleList(2,MenuId,0,"","")
%>
			<tr>
				<td height="30" colspan="20" align="center">
					<button type="submit" class="button" name="Enter" id="Enter">排 序</button>
				</td>
			</tr>
		</table>
		</form>
		<div id="rightBottom">
		
		</div>
<%
End Sub

'==============================
'函 数 名：ModuleOrderDo
'作	用：执行菜单项排序
'==============================
Sub ModuleOrderDo()
	Sqlstr="Select Fk_Module_Id,Fk_Module_Name,Fk_Module_Order From [Fk_Module] Where Fk_Module_Menu="&MenuId&" Order By Fk_Module_Id Desc"
	Rs.Open Sqlstr,Conn,1,3
	Application.Lock()
	While Not Rs.Eof
		Fk_Module_Order=Trim(Request.Form("Fk_Module_Order"&Rs("Fk_Module_Id")))
		Call FKFun.JsonNum(Fk_Module_Order,Rs("Fk_Module_Name")&"模块的序号不是数字，排序序号必须是有效数字！")
		Rs("Fk_Module_Order")=Fk_Module_Order
		Rs.Update()
		Rs.MoveNext
	Wend
	Application.UnLock()
	Rs.Close
	Call FKFun.JsonSuccess()
End Sub

'==============================
'函 数 名：GetModelName
'作	用：获取模型名称
'==============================
Function GetModelName(gId)
	Dim tempRs
	Set tempRs=Server.Createobject("Adodb.RecordSet")
	Sqlstr="Select Fk_Model_Name From [Fk_Model] Where Fk_Model_Id=" & gId
	tempRs.Open Sqlstr,Conn,1,1
	If Not tempRs.Eof Then
		GetModelName=tempRs("Fk_Model_Name")
	Else
		GetModelName="<span style='color:red;'>[未知模型]</span>"
	End If
	tempRs.Close
End Function









'==========================================
'函 数 名：ModuleBatAddForm()
'作	用：批量添加名称表单
'==========================================
Sub ModuleBatAddForm()
	Sqlstr="Select Fk_Template_Name From [Fk_Template] Where Not (Fk_Template_Name Like 'm_%%' Or Fk_Template_Name Like '%%/m_%%') Order By Fk_Template_Id Asc"
	Rs.Open Sqlstr,Conn,1,1
	While Not Rs.Eof
		If Instr(Rs("Fk_Template_Name"),"/")>0 Then
			TempArr=Split(Rs("Fk_Template_Name"),"/")
			Temp=","&TempArr(UBound(TempArr))&","
		Else
			Temp=","&Rs("Fk_Template_Name")&","
		End If
		If Instr(DisTemplate,Temp)=0 Then
			If TemplateSelect="" Then
				TemplateSelect=Rs("Fk_Template_Name")
			Else
				TemplateSelect=TemplateSelect&",,"&Rs("Fk_Template_Name")
			End If
		End If
		Rs.MoveNext
	Wend
	Rs.Close
%>
<form id="ModuleAdd" name="ModuleAdd" method="post" target="ajax" action="Module.asp?Type=30&Menu=<%=MenuId%>" showStr="新菜单项批量添加成功！" refreshUrl="Module.asp?Type=1&Menu=<%=MenuId%>" boxClose="1">
<div id="title">批量添加新菜单项</div>
<div id="box">
	<table width="90%" border="1" bordercolor="#CCCCCC" align="center" cellpadding="0" cellspacing="0">
		<tr>
			<td width="24%" height="30" align="right">名称：</td>
			<td width="76%">
				<textarea name="Fk_Module_Name" style="width:340px;height:120px;" class="textarea" id="Fk_Module_Name"></textarea>
				<span class="help" title="<p><b>一行一个</b>菜单项名称，请输入1-50个字符（两个字符为一个汉字）。</p>"></span>
			</td>
		</tr>
		<tr>
			<td height="30" align="right">副标题：</td>
			<td>&nbsp;<input name="Fk_Module_SubTitle" type="text" class="input" id="Fk_Module_SubTitle" /><span class="help" title="<p>菜单项副标题，请输入1-50个字符（1个汉字为2个字符）。</p>"></span></td>
		</tr>
		<tr>
			<td height="30" align="right">关键字：</td>
			<td>&nbsp;<input name="Fk_Module_Keyword" type="text" class="input" id="Fk_Module_Keyword" size="50" /><span class="help" title="<p>多个关键字用英文逗号隔开，用于页面meta的keywords，请输入1-255个字符（两个字符为一个汉字）。</p>"></span></td>
		</tr>
		<tr>
			<td height="30" align="right">描述：</td>
			<td>&nbsp;<input name="Fk_Module_Description" type="text" class="input" id="Fk_Module_Description" size="50" /><span class="help" title="<p>请输入菜单项的一段文字描述，用于页面meta的description，请输入1-255个字符（两个字符为一个汉字）。</p>"></span></td>
		</tr>
		<tr>
			<td height="30" align="right">图片：</td>
			<td>&nbsp;<input name="Fk_Module_Pic" type="text" class="input" id="Fk_Module_Pic" size="30" /> <input type="button" class="picList_btn btnUpload" name="btnPic" id="btnPic" value="上传图片" callback="2" inputId="Fk_Module_Pic" fileAction="Upload.asp?immediate=3" /><span class="help" title="<p>上传菜单图片</p>"></span></td>
		</tr>
		<tr class="dis">
			<td height="30" align="right">绝对路径：</td>
			<td>&nbsp;<input name="Fk_Module_FixedUrl" type="text" class="input" id="Fk_Module_FixedUrl" size="30" /> *设定后无法修改<span class="help" title="<p>绝对路径，请输入1-255个字符，设定后将忽略目录设置，内容模块和带分页的留言模块必须要是目录。</p>"></span></td>
		</tr>
		<tr class="dis">
			<td height="30" align="right">目录：</td>
			<td>&nbsp;<input name="Fk_Module_Dir" type="text" class="input" id="Fk_Module_Dir" size="30" /> *设定后无法修改<span class="help" title="<p>目录名，请输入1-50个字符，必须是字母或数字。</p>"></span></td>
		</tr>
		<tr>
			<td height="30" align="right">分级：</td>
			<td>&nbsp;<select class="input" name="Fk_Module_Level" id="Fk_Module_Level">
					<option value="0">一级菜单项</option>
<%
	Call GetModuleList(1,MenuId,0,"","")
%>
				</select><span class="help" title="<p>菜单项分级。</p>"></span>
			</td>
		</tr>
		<tr>
			<td height="30" align="right">类型：</td>
			<td>&nbsp;<select class="input" name="Fk_Module_Type" id="Fk_Module_Type" onchange="ModuleTypeChange($(this).val());">
					<option value="0">静态模块</option>
					<option value="1">内容模块</option>
					<option value="2">信息模块</option>
					<option value="3">留言模块</option>
				</select><span class="help" title="<p>菜单项类型。</p>"></span>
			</td>
		</tr>
		<tr class="m_Url dis">
			<td height="30" align="right">跳转模式：</td>
			<td>&nbsp;<input name="Fk_Module_LinkType" type="radio" class="input" id="Fk_Module_LinkType" value="0" checked="checked" />直接跳转
				<input name="Fk_Module_LinkType" type="radio" class="input" id="Fk_Module_LinkType" value="1" />页面跳转<span class="help" title="<p>设置跳转链接的方式，如果选择“直接跳转”，则会直接输出链接到页面，可以做外链；如选择“页面跳转”，会生成一个页面，并在通过页面内代码跳转。</p>"></span>
			</td>
		</tr>
		<tr class="m_Url dis">
			<td height="30" align="right">跳转链接：</td>
			<td>&nbsp;<input name="Fk_Module_Link" type="text" class="input" id="Fk_Module_Link" size="50" /> <span class="help" title="<p>设置一个链接，链接请输入1-255个字符，支持相对链接。</p>"></span></td>
		</tr>
		<tr id="m_Template">
			<td height="30" align="right">显示模板：</td>
			<td>&nbsp;<select class="input" name="Fk_Module_Template" id="Fk_Module_Template">
					<option value="">默认模板</option>
<%
	TempArr=Split(TemplateSelect,",,")
	For Each Temp In TempArr
%>
					<option value="<%=Temp%>"><%=Temp%></option>
<%
	Next
%>
				</select><span class="help" title="<p>选择显示模板。</p>"></span>
			</td>
		</tr>
		<tr class="m_List dis">
			<td height="30" align="right">子内容模板：</td>
			<td>&nbsp;<select class="input" name="Fk_Module_LowTemplate" id="Fk_Module_LowTemplate">
					<option value="">默认模板</option>
<%
	TempArr=Split(TemplateSelect,",,")
	For Each Temp In TempArr
%>
					<option value="<%=Temp%>"><%=Temp%></option>
<%
	Next
%>
				</select><span class="help" title="<p>选择子内容模板。</p>"></span>
			</td>
		</tr>
		<tr id="m_Model" class="dis">
			<td height="30" align="right">模型：</td>
			<td>&nbsp;<select class="input" name="Fk_Module_Model" id="Fk_Module_Model">
					<option value="0">请先选择类型</option>
				</select><span class="help" title="<p>菜单项类型。</p>"></span>
			</td>
		</tr>
		<tr id="m_GBookReply" class="dis">
			<td height="30" align="right">留言分页输出：</td>
			<td>&nbsp;<select class="input" name="Fk_Module_GBookReply" id="Fk_Module_GBookReply">
					<option value="0">所有记录</option>
					<option value="1">已回复记录</option>
					<option value="2">未回复记录</option>
					<option value="3">不在前台显示留言</option>
				</select><span class="help" title="<p>留言分页输出记录类型。</p>"></span>
			</td>
		</tr>
		<tr class="m_List dis">
			<td height="30" align="right">每页条数：</td>
			<td>&nbsp;<select class="input" name="Fk_Module_PageSize" id="Fk_Module_PageSize">
					<option value="0">按站点设置配置条数</option>
<%
	For i=1 To 50
%>
					<option value="<%=i%>"><%=i%>条</option>
<%
	Next
%>
				</select><span class="help" title="<p>菜单项类型。</p>"></span>
			</td>
		</tr>
<%
	Sqlstr="Select Fk_Field_Id,Fk_Field_Name,Fk_Field_Remark,Fk_Field_Label,Fk_Field_Help,Fk_Field_Type,Fk_Field_Option From [Fk_Field] Where Fk_Field_Use=2 Order By Fk_Field_Id Asc"
	Rs.Open Sqlstr,Conn,1,1
	While Not Rs.Eof
		Select Case Rs("Fk_Field_Type")
			Case 0
				Call FKFun.GetField_Input(Rs("Fk_Field_Name"),Rs("Fk_Field_Label"),Rs("Fk_Field_Help"),"")
			Case 1
				Call FKFun.GetField_Editor(Rs("Fk_Field_Name"),Rs("Fk_Field_Label"),Rs("Fk_Field_Help"),"")
			Case 2
				Call FKFun.GetField_Upload(Rs("Fk_Field_Name"),Rs("Fk_Field_Label"),Rs("Fk_Field_Help"),"")
			Case 3
				Call FKFun.GetField_Select(Rs("Fk_Field_Name"),Rs("Fk_Field_Label"),Rs("Fk_Field_Help"),Rs("Fk_Field_Option"),"")
		End Select
		Rs.MoveNext
	Wend
	Rs.Close
%>
		<tr>
			<td height="30" align="right">菜单是否显示：</td>
			<td>&nbsp;
				<label><input type="radio" name="Fk_Module_MenuShow" class="input" id="Fk_Module_MenuShow" value="0" />不显示</label>
				<label><input type="radio" name="Fk_Module_MenuShow" class="input" id="Fk_Module_MenuShow" value="1" checked="checked" />显示</label><span class="help" title="<p>配置该菜单项是否在菜单输出里显示。</p>"></span>
			</td>
		</tr>
		<tr>
			<td height="30" align="right">是否可用：</td>
			<td>&nbsp;
				<label><input type="radio" name="Fk_Module_State" class="input" id="Fk_Module_State" value="0" />不可用</label>
				<label><input type="radio" name="Fk_Module_State" class="input" id="Fk_Module_State" value="1" checked="checked" />可用</label> <span class="help" title="<p>配置该菜单项是否可用。</p>"></span>
			</td>
		</tr>
<%
	If Fk_Module_IsIndexShow="1" Then
%>
		<tr>
			<td height="30" align="right">菜单首页：</td>
			<td>&nbsp;
				<label><input type="radio" name="Fk_Module_IsIndex" class="input" id="Fk_Module_IsIndex" value="0" />否</label>
				<label><input type="radio" name="Fk_Module_IsIndex" class="input" id="Fk_Module_IsIndex" value="1" checked="checked" />是</label><span class="help" title="<p>菜单首页。</p>"></span>
			</td>
		</tr>
<%
	End If
%>
	</table>
</div>
<div id="bottom">
	<input type="submit" class="button" name="Enter" id="Enter" value="添 加" />
	<input type="button" onclick="dialogClose();" class="button" name="button" id="button" value="关 闭" />
</div>
</form>
<%
End Sub




'==============================
'函 数 名：ModuleBatAddDo
'作	用：执行添加
'==============================
Sub ModuleBatAddDo()
	Fk_Module_Name=FKFun.HTMLEncode(Trim(Request.Form("Fk_Module_Name")))
	Fk_Module_SubTitle=FKFun.HTMLEncode(Trim(Request.Form("Fk_Module_SubTitle")))
	Fk_Module_Keyword=FKFun.HTMLEncode(Trim(Request.Form("Fk_Module_Keyword")))
	Fk_Module_Description=FKFun.HTMLEncode(Trim(Request.Form("Fk_Module_Description")))
	Fk_Module_Pic=FKFun.HTMLEncode(Trim(Request.Form("Fk_Module_Pic")))
	Fk_Module_FixedUrl=FKFun.HTMLEncode(Trim(Request.Form("Fk_Module_FixedUrl")))
	Fk_Module_Dir=Server.URLEncode(FKFun.HTMLEncode(Trim(Request.Form("Fk_Module_Dir"))))
	Fk_Module_Template=FKFun.HTMLEncode(Trim(Request.Form("Fk_Module_Template")))
	Fk_Module_LowTemplate=FKFun.HTMLEncode(Trim(Request.Form("Fk_Module_LowTemplate")))
	Fk_Module_Link=FKFun.HTMLEncode(Trim(Request.Form("Fk_Module_Link")))

	Fk_Module_Type=Trim(Request.Form("Fk_Module_Type"))
	Fk_Module_PageSize=Trim(Request.Form("Fk_Module_PageSize"))
	Fk_Module_Level=Trim(Request.Form("Fk_Module_Level"))
	Fk_Module_Model=Trim(Request.Form("Fk_Module_Model"))
	Fk_Module_GBookReply=Trim(Request.Form("Fk_Module_GBookReply"))
	Fk_Module_State=Trim(Request.Form("Fk_Module_State"))
	Fk_Module_MenuShow=Trim(Request.Form("Fk_Module_MenuShow"))
	Fk_Module_LinkType=Trim(Request.Form("Fk_Module_LinkType"))
	Fk_Module_IsIndex=Trim(Request.Form("Fk_Module_IsIndex"))
	
	Call FKFun.JsonString(Fk_Module_Name,1,5000,0,"请输入名称！","名称不能大于50个字符！")
	Call FKFun.JsonString(Fk_Module_SubTitle,1,50,2,"请输入副标题！","副标题不能大于50个字符！")
	Call FKFun.JsonString(Fk_Module_Keyword,1,255,2,"请输入关键字！","关键字不能大于255个字符！")
	Call FKFun.JsonString(Fk_Module_Description,1,255,2,"请输入描述！","描述不能大于255个字符！")
	Call FKFun.JsonString(Fk_Module_Pic,1,50,2,"请上传图片！","图片不能大于50个字符！")
	Call FKFun.JsonString(Fk_Module_FixedUrl,1,255,2,"请输入绝对路径！","绝对路径不能大于255个字符！")
	Call FKFun.JsonString(Fk_Module_Dir,1,50,2,"请输入目录名！","目录名不能大于50个字符！")
	Call FKFun.JsonNum(Fk_Module_Level,"请选择菜单项分级！")
	Call FKFun.JsonNum(Fk_Module_State,"请选择菜单项是否可用！")
	Call FKFun.JsonNum(Fk_Module_MenuShow,"请选择菜单项是否在菜单上显示！")
	Call FKFun.JsonNum(Fk_Module_Type,"请选择菜单项类型！")
	If Instr("123",Fk_Module_Type)>0 Then
		Call FKFun.JsonNum(Fk_Module_Model,"请选择菜单项模型！")
	Else
		Fk_Module_Model=0
	End If
	If Instr("13",Fk_Module_Type)>0 Then
		Call FKFun.JsonNum(Fk_Module_PageSize,"请选择每页条数！")
	Else
		Fk_Module_PageSize=0
	End If
	If Instr("3",Fk_Module_Type)>0 Then
		Call FKFun.JsonNum(Fk_Module_GBookReply,"请选择留言分页输出类型！")
	Else
		Fk_Module_GBookReply=0
	End If
	


	
	'分级与链接读取
	If Fk_Module_Level>0 Then
		Sqlstr="Select Fk_Module_LevelList,Fk_Module_Url From [Fk_Module] Where Fk_Module_Id="&Fk_Module_Level&" And Fk_Module_Menu="&MenuId&""
		Rs.Open Sqlstr,Conn,1,1
		If Not Rs.Eof Then
			ModuleLevelList=Rs("Fk_Module_LevelList")
			ModuleUrl=Rs("Fk_Module_Url")
		End If
		Rs.Close
	Else
		ModuleLevelList="0"
		ModuleUrl=""
	End If
	If Fk_Module_Level=0 Then
		Fk_Module_LevelList="0"
	Else
		If ModuleLevelList="0" Then
			Fk_Module_LevelList=","&Fk_Module_Level&","
		Else
			Fk_Module_LevelList=ModuleLevelList&Fk_Module_Level&","
		End If
	End If
	
	If Fk_Module_Dir="" Then
		Fk_Module_Url=ModuleUrl
	Else
		If MenuDir<>"" Then
			Fk_Module_Url=ModuleUrl&MenuDir&"/"&Fk_Module_Dir&"/"
		Else
			Fk_Module_Url=ModuleUrl&Fk_Module_Dir&"/"
		End If
	End If
	
	'菜单首页链接
	'If Fk_Module_IsIndex=1 Then
	'	Fk_Module_Url=MenuDir&"/"
	'End If
	'Response.Write MenuDir & "--Module_Dir[" & Fk_Module_Dir & "] ModuleUrl [" & ModuleUrl & "]--Module_Url " & Fk_Module_Url
	'Response.End

	Fk_Module_IsIndex = 0
	
	Dim title_arr,err_arr
	If InStr(Fk_Module_Name,"<br")>0 Then
		title_arr = Split(Fk_Module_Name,"<br />")
	ElseIf InStr(Fk_Module_Name,"<p")>0 Then
		title_arr = Split(Fk_Module_Name,"<p></p>")
	Else
		title_arr = Split(Fk_Module_Name,vbcrlf)
	End If

	Dim title_fk,Fk_Module_Url_xx
	Fk_Module_Url_xx = ""
	err_arr = ""



	For i=0 To UBound(title_arr)
		title_fk = title_arr(i)
		title_fk = Trim(title_fk)


		If title_fk<>"" Then
	'===============================================
		
	Sqlstr="Select Fk_Module_Id,Fk_Module_Name,Fk_Module_SubTitle,Fk_Module_Keyword,Fk_Module_Description,Fk_Module_Pic,Fk_Module_About,Fk_Module_Content,Fk_Module_Type,Fk_Module_FixedUrl,Fk_Module_Dir,Fk_Module_Url,Fk_Module_PageSize,Fk_Module_Model,Fk_Module_GBookReply,Fk_Module_IsIndex,Fk_Module_Menu,Fk_Module_Level,Fk_Module_Order,Fk_Module_LevelList,Fk_Module_Click,Fk_Module_Template,Fk_Module_LowTemplate,Fk_Module_LinkType,Fk_Module_Link,Fk_Module_State,Fk_Module_MenuShow,Fk_Module_Field,Fk_Module_ToHtml,Fk_Module_Admin,Fk_Module_Ip,Fk_Module_Time From [Fk_Module] Where Fk_Module_Name='"&title_fk&"' And Fk_Module_Menu="&MenuId&" And Fk_Module_Level="&Fk_Module_Level&""
	Rs.Open Sqlstr,Conn,1,3
	If Rs.Eof Then
		Application.Lock()
		
		Rs.AddNew()
		Rs("Fk_Module_Name")=title_fk
		Rs("Fk_Module_SubTitle")=Fk_Module_SubTitle
		Rs("Fk_Module_Keyword")=Fk_Module_Keyword
		Rs("Fk_Module_Description")=Fk_Module_Description
		Rs("Fk_Module_Pic")=Fk_Module_Pic
		Rs("Fk_Module_About")=""
		Rs("Fk_Module_Content")=""
		Rs("Fk_Module_Type")=Fk_Module_Type
		Rs("Fk_Module_FixedUrl")=Fk_Module_FixedUrl
		Rs("Fk_Module_Dir")=Fk_Module_Dir
		Rs("Fk_Module_Url")=Fk_Module_Url
		Rs("Fk_Module_PageSize")=Fk_Module_PageSize
		Rs("Fk_Module_Model")=Fk_Module_Model
		Rs("Fk_Module_GBookReply")=Fk_Module_GBookReply
		Rs("Fk_Module_IsIndex")=Fk_Module_IsIndex
		Rs("Fk_Module_Menu")=MenuId
		Rs("Fk_Module_Level")=Fk_Module_Level
		Rs("Fk_Module_Order")=0
		Rs("Fk_Module_LevelList")=Fk_Module_LevelList
		Rs("Fk_Module_Click")=0
		Rs("Fk_Module_Template")=Fk_Module_Template
		Rs("Fk_Module_LowTemplate")=Fk_Module_LowTemplate
		Rs("Fk_Module_LinkType")=Fk_Module_LinkType
		Rs("Fk_Module_Link")=Fk_Module_Link
		Rs("Fk_Module_State")=Fk_Module_State
		Rs("Fk_Module_MenuShow")=Fk_Module_MenuShow
		Rs("Fk_Module_Field")=Fk_Module_Field
		Rs("Fk_Module_ToHtml")=1
		Rs("Fk_Module_Admin")=UId
		Rs("Fk_Module_Ip")=Request.ServerVariables("REMOTE_ADDR")
		Rs("Fk_Module_Time")=Now()
		Rs.Update()

		If Fk_Module_Dir="" And Fk_Module_IsIndex=0 Then
			Rs.Close
			Sqlstr="Select Fk_Module_Id,Fk_Module_Url From [Fk_Module] Where Fk_Module_Name='"& title_fk &"' And Fk_Module_Menu="&MenuId&""
			Rs.Open Sqlstr,Conn,1,3
			If Not Rs.Eof Then
				If MenuDir<>"" Then
					MenuDir=MenuDir&"/"
				End If
				Fk_Module_Url_xx =Fk_Module_Url&MenuDir&"Col"&Rs("Fk_Module_Id")&"/"
				Fk_Module_Url_xx=Replace(Fk_Module_Url_xx,"//","/")
				Rs("Fk_Module_Url")=Fk_Module_Url_xx
				Rs.Update()
			End If
		End If
		
		If Fk_Module_Level>0 Then
			Sqlstr="Update [Fk_Module] Set Fk_Module_SubColCount=Fk_Module_SubColCount+1 Where Fk_Module_Id=" & Fk_Module_Level
			Conn.Execute(Sqlstr)
			If Fk_Site_Html=2 Then
				Sqlstr="Update [Fk_Module] Set Fk_Module_ToHtml=1 Where Fk_Module_Id In (0"&Fk_Module_LevelList&"0)"
				Conn.Execute(Sqlstr)
			End If
		End If

		
		Application.UnLock()
		'Call FKFun.JsonSuccess()
	Else
		'Rs.Close
		'Call FKFun.JsonError("该菜单项已存在！")
		err_arr = err_arr & title_fk & ","
	End If
		
	'==========================================================

			Rs.Close
			
		End If
	Next
	'Rs.Close

	If err_arr="" Then
		Call FKFun.JsonSuccess()
	Else
		Call FKFun.JsonError("菜单项["&err_arr&"]已存在！")
	End If
	
End Sub








Call FKFun.PageEnd()
%>
